CALCULATING DEW POINT FROM RH and AIR TEMPERATURE

Introduction

Dew point can be measured directly, to a high degree of accuracy, using traditional devices such as cooled mirror hygrometers, etc.

However, such devices are often very expensive, require regular maintenance and may need air pumps. They are also heavy on power consumption.

An alternative method, described in this Technical Note, uses relatively inexpensive RH and Temperature sensors, in conjunction with a Campbell Scientific datalogger, to calculate dew point. While end results may not be quite as accurate as traditional dedicated devices, they are acceptable for a wide range of applications.

Calculating Dew Point

Dew point temperature can be calculated by Campbell Scientific dataloggers as follows:

1. Measure the relative humidity (RH) and air temperature (T_a; units °C).
2. Compute the saturation vapour pressure (S_{vp}; units kPa) using Instruction 56.
3. Compute the vapour pressure (V_p; units kPa) from $V_p = RH * S_{vp} / 100$).
4. Compute the dew point (Td; units °C) from the inverse of a version of Tetens' equation, optimised for dewpoints in the range -35 to 50°C:

$$T_d = (C_3 * \ln(V_p / C_1)) / (C_2 - \ln(V_p / C_1))$$

where:

$C_1 = 0.61078$
$C_2 = 17.558$
$C_3 = 241.88$

Error in the Estimation of Dew Point

Teten's equation is an approximation of the true variation of saturated vapour pressure as a function of temperature. However, the errors in using the inverted form of the equation result in dew point errors much less than 0.1°C.

The largest component of error, in reality, comes from errors in the absolute calibration of the temperature and RH sensor.

Figure 1 shows how dew point varies as a function of temperature and humidity. It can be seen that the response is non-linear with respect to both variables. Errors in the measurement of RH and temperature thus form a complex function in relation to the resultant error in estimated dew point. In practise, the effect of errors in the calibration of air temperature can be taken to translate to an equivalent error in dew point, e.g. if the air temperature sensor is 0.2°C high, then the estimated dew point is approximately 0.2°C high. Figure 2 shows the errors in dew point as a function of a 'worst case' 5% error in the calibration of the RH sensor.

For sensors installed in the field there are additional errors associated with exposure of the sensor, e.g. sensors in unaspirated shields get slightly warmer than true air temperature in conditions of low wind speeds and high solar radiation. However, if the RH and air temperature sensors are installed in the same shield and are thus exposed identically, the estimate of dew point is not subject to the same error as the measurement of air temperature would be. This is because the temperature sensor will measure the actual temperature of the RH sensor, which is what is required for the derivation of air vapour pressure and thereby dew point.



Figure 1. Dew Point Temperature over the RH Range for Selected Air Temperatures



Figure 2. Effect of RH Errors on Calculated Dew Point (±5 RH Unit Error at Three Air Temperatures)

Program Example

```
;{CR10X}
;Program: Demonstration of DewPoint
;calculation
;This example uses an HMP45C Relative
;Humidity and Temperature probe/

*Table 1 Program
  01: 60   Execution Interval (seconds)
```

;First turn on power to the probe

```
1:  Do (P86)
 1: 41       Set Port 1 High
```

;Use Instruction 22 to force a 1 second settling delay

```
2:  Excitation with Delay (P22)
 1: 1        Ex Channel
 2: 0        Delay W/Ex (units = 0.01 sec)
 3: 100      Delay After Ex (units = 0.01 sec)
 4: 0        mV Excitation
```

;Measure RH and Temperature mV and multiply ;readings by 0.1

```
3:  Volt (SE) (P1)
 1: 2        Reps
 2: 5        2500 mV Slow Range
 3: 1        SE Channel
 4: 1        Loc [ RH        ]
 5: 0.1      Mult
 6: 0.0      Offset
```

;Turn off power to the probe

```
4:  Do (P86)
 1: 51       Set Port 1 Low
```

;Subtract 40 from Temperature to scale to Celsius

```
5:  Z=X+F (P34)
 1: 2        X Loc [ Air_Temp ]
 2: -40      F
 3: 2        Z Loc [ Air_Temp ]

6:  Saturation Vapor Pressure (P56)
 1: 2        Temperature Loc [ Air_Temp ]
 2: 4        Loc [ Sat_VP    ]
```

;Now calculate Vapour pressure using
*;VP = RH * Sat_VP / 100*
;This equation can be entered directly for Edlog 6+.
;Instructions 7 - 8 show the instructions required for
;older versions of Edlog or keyboard entry.

```
7:  Z=X*Y (P36)
 1: 4        X Loc [ Sat_VP    ]
 2: 1        Y Loc [ RH        ]
 3: 7        Z Loc [ WORK_1     ]
```

;Multiply by 0.01 (equiv. to dividing by 100)

```
8:  Z=X*F (P37)
 1: 7        X Loc [ WORK_1     ]
 2: .01      F
 3: 5        Z Loc [ VP        ]
```

;Now estimate dew point using the equation:
*;Dew_Temp = 241.88 * ln(VP/0.61078) / (17.558 -*
;ln(VP/0.61078))
;This equation can be entered directly with Edlog 6+.
;Instructions 9 - 14 show the instructions required for
;older versions of Edlog or keyboard entry

;Multiply VP by 1/0.61078 (= 1.6373)

```
9:  Z=X*F (P37)
 1: 5        X Loc [ VP        ]
 2: 1.6373   F
 3: 6        Z Loc [ WORK_R     ]

10:  Z=LN(X) (P40)
 1: 6        X Loc [ WORK_R     ]
 2: 6        Z Loc [ WORK_R     ]


11:  Z=X*F (P37)
 1: 6        X Loc [ WORK_R     ]
 2: 241.88   F
 3: 7        Z Loc [ WORK_1     ]

12  Z=F (P30)
 1: 17.558   F
 2: 0        Exponent of 10
 3: 8        Z Loc [ WORK_2     ]

13  Z=X-Y (P35)
 1: 8        X Loc [ WORK_2     ]
 2: 6        Y Loc [ WORK_R     ]
 3: 8        Z Loc [ WORK_2     ]

14  Z=X/Y (P38)
 1: 7        X Loc [ WORK_1     ]
 2: 8        Y Loc [ WORK_2     ]
 3: 3        Z Loc [ DEW_TEMP  ]
```

;And now, as an example, store the time and hourly
;average

```
15  If time is (P92)
 1: 0        Minutes (Seconds --) into a
 2: 60       Interval (same units as above)
 3: 10       Set Output Flag High

16  Real Time (P77)
 1: 110      Day,Hour/Minute

17  Average (P71)
 1: 1        Reps
 2: 3        Loc [ DEW_TEMP  ]

*Table 2 Program
  02: 0      Execution Interval (seconds)

*Table 3 Subroutines

End Program
```

Input Locations

```
1  RH        1 1 1
2  Air_temp  1 1 1
3  DEW_TEMP  1 1 1
4  Sat_VP    1 1 1
5  VP        1 1 1
6  WORK_R     1 3 2
7  WORK_1     1 2 2
8  WORK_2     1 2 2
9 _ _________ 0 0 0
10 _________ 0 0 0
11 _________ 0 0 0
12 _________ 0 0 0
13 _________ 0 0 0
14 _________ 0 0 0
15 _________ 0 0 0
16 _________ 0 0 0
27 ______ 0 0 0
28 ______ 0 0 0

-Program Security-
00
0000
-Mod
-Mode 4-
-Final Storage Area 2-
0
```